

EXCEL2010

# 高效办公助手

常用函数应用

大纲牙甜甜整理  
2012-4-12

# 目 录

单条件求和——SUMIF 函数.....	2
符合多个条件的数据求和——SUMIFS 函数.....	2
满足多重条件的单元格数目——COUNTIFS 函数.....	2
不显示计算结果为 0 值的项——AND 函数、IF 函数.....	3
判断单元格是否为空——ISBLANK 函数.....	4
名次排位函数——RANK 函数.....	5
提取字符串中的某些字符函数——LEFT, RIGHT, MID, .....	7
插入当前时间函数——NOW 函数.....	8
计算两个日期期间的天数——DATEVALUE 函数.....	9
分解时间函数——YEAR 函数.....	9
计算员工工龄函数——YEAR 函数.....	11
去尾取整函数——INT 函数.....	11
向下舍数字函数——ROUNDDOWN 函数.....	11
余数函数——MOD 函数.....	12
四舍五入函数——ROUND 函数.....	13
查找函数.....	13
寻找并参照函数——vlookup 函数（只查找第一次出现的）.....	13
VLOOKUP 函数的使用方法（入门级）.....	14
VLOOKUP 函数的使用方法（初级篇）.....	16
VLOOKUP 函数的使用方法（进阶篇）.....	18
VLOOKUP 函数的使用方法(高级篇).....	21
返回指定行、列或单元格的值——INDEX 函数.....	24
返回匹配值的位置函数——MATCH 函数（只查找第一次出现的）.....	26
INDEX 和 MATCH 函数组合应用（相当于 vlookup 函数）.....	27
引用函数——OFFSET.....	31
OFFSET 函数的用法及应用实战讲解（上）.....	34
OFFSET 函数的经典应用（下）.....	37

## 单条件求和——SUMIF 函数

如要统计 C 列中的数据,要求统计条件是 B 列中数据为“条件一”。并将结果放在 C6 单元格中,我们只要在 C6 单元格中输入公式“=SUMIF (B2: B5, “条件一”, C2: C5)”即完成这一统计。

## 符合多个条件的数据求和——SUMIFS 函数

统计 5 班中语文名次、数学名次均位于前 20 名同学的总分之和。在相应单元格内输入公式 “=SUMIFS (M2:M80, A2:A80, “5”, D2:D80, “<=20”, F2:F80, “<=20”)”, 按下回车键, 就一切 OK。如图所示。

	A	B	C	D	E	F	G	H	I	J	K	L	M
1	班级	姓名	语文	名次	数学	名次	英语	名次	理论	名次	实践	名次	总分
2	5班	梅	79	1	82	12	91	1	190	3	180	1	639
3	5班	萌	79	1	70	49	86	8	153	41	115	50	602
4	6班	永	62	44	88	5	91	1	184	9	169	5	653
5	6班	珍	73	8	82	12	84	12	185	8	165	9	629
6	6班	敬	69	16	91	2	83	14	180	13	166	7	634
7	5班	洋	75	3	77	28	86	8	126	67	88	71	558
8	6班	坤	68	19	77	28	85	10	186	6	163	13	642
9	6班	洁	71	14	72	42	77	29	186	8	171	8	668
10	6班	玲	68	19	75	34	82	17	188	10	166	7	654
11	6班	霞	59	57	88	5	78	26	188	4	159	15	664

M	N	O	P	Q
总分				
639				
602			3776	=SUMIFS (M2:M80, A2:A80, “5班”, D2:D80,
653				

## 满足多重条件的单元格数目——COUNTIFS 函数

COUNTIFS 函数计算某个区域中满足多重条件的单元格数目。

要统计满足多个条件的学生人数,可以使用 COUNTIFS 函数统计满足多个条件的学生人数。如上图所示,在 C8 单元格输入公式:

=COUNTIFS (B2:B5, “>=60”, C2:C5, “>=60”)

=COUNTIFS (B2:B5, “>=60”, C2:C5, “>=60”)				
Excel伙伴天下.xlsx				
	A	B	C	D
1	姓名	语文	数学	
2	张三	90	80	
3	李四	70	85	
4	陈平	60	98	
5	刘洋	58	76	
6				
7				
8	统计人数:		3	

按回车键即可统计满足多个条件的学生人数。返回语文和数学成绩都及格的人数。

相关说明：

COUNTIFS 函数语法：COUNTIFS(range1, criterial, range2, criteria2...)

range1, range2, ...：是计算关联条件的 1 至 127 个区域。每个区域中的单元格必须是数字或包含数字的名称、数组或引用。空值和文本值会被忽略。

criterial, criteria2, ...：是数字、表达式、单元格引用或文本形式的 1 至 127 个条件，用于定义要对哪些单元格进行计算。例如，条件可以表示为 32、“32”、“>32”、“apples”或 B4。

仅当区域中的每一单元格满足为其指定的所有相应条件时才对其进行计算。

可以在条件中使用通配符、问号 (?) 和星号 (\*)。问号匹配任意单个字符；星号匹配任意一串字符。如果要查找实际的问号或星号，请在该字符前键入波形符 (~)。

补充：

大钢牙甜甜	个	1
w	呃	2
e	钢	3
1	大	4
	大钢牙甜甜	5
w	飞	6

区域中“大钢牙甜甜”出现次数：，=COUNTIF(A1:C6, “大钢牙甜甜”)

返回值为：2

## 不显示计算结果为 0 值的项——AND 函数、IF 函数

在 Excel 中，使用公式进行填充运算时，当引用单元格都没有输入数值时，结果单元格会显示 0 值，如果不显示计算结果为 0 值的项，可使用 AND 函数不显示计算结果为 0 值的项。

D2

fx

=B2+C2

Excel伙伴天下.xlsx

	A	B	C	D
1	姓名	产品1销售量	产品2销售量	总销售额
2	张三			0
3	李四	30	10	40
4	孙海	20		20
5	总计:	50	10	60

如上图所示，当 B2 和 C2 单元格同时为空值时，D2 单元格公式的计算结果显示为“0”。如果不显示计算结果为 0 值的项可将 D2 单元格的公式改为：

=IF (AND (B2="", C2=""), "", B2+C2)

D2		=IF (AND (B2="", C2=""), "", B2+C2)			
Excel伙伴天下.xlsx					
	A	B	C	D	E
1	姓名	产品1销售量	产品2销售量	总销售额	
2	张三				
3	李四	30	10	40	
4	孙海	20		20	
5	总 计:	50	10	60	

如上图所示，当 B2 和 C2 单元格同时为空值时，D2 单元格不显示 0 值。其它的公式可使用 Excel 自动填充进行公式填充。

Excel2007 使用 AND 函数可判断不显示计算结果为 0 值的项。

#### 相关说明：

- AND 函数语法：AND(logical1, logical2, ...)
- 所有参数的逻辑值为真时，返回 TRUE；只要一个参数的逻辑值为假，即返回 FALSE。
- 参数必须是逻辑值 TRUE 或 FALSE，或者包含逻辑值的数组或引用。
- 如果数组或引用参数中包含文本或空白单元格，则这些值将被忽略。
- 如果指定的单元格区域内包括非逻辑值，则 AND 将返回错误值 #VALUE!。

## 判断单元格是否为空——ISBLANK 函数

第一，isblank 函数的用法判断单元格是否为空

Excel 中 isblank 函数的用法是判断单元格是否为空。

下图所示：A1 单元格没有输入任何东西，A2 单元格输入：“”。

我们在 C1 单元格输入公式：=IF(ISBLANK(A1), TRUE, FALSE)，公式的意思是用 ISBLANK(A1) 做为 IF 条件，如果单元格为空，则返回 TRUE，反之返回 FALSE。

C1	A	B	C
1		真空	TRUE
2		空字符串	FALSE
3			

解释：isblank 函数的用法是判别该单元格是否为空单元格（真空），而 A2 单元格中是“”为假空而非真空单元格，所以会返回 FALSE。

第二，isblank 函数的用法实例

从帮助文件我们得知 isblank 函数的用法是检验指定值并返回 TRUE 或 FALSE。例如，如果参数 value 引用的是空单元格，ISBLANK 函数返回逻辑值 TRUE；否则，返回 FALSE。

下面截图根据 AB 列成绩，如果两科成绩都不为空，就求两科的平均分，如果任意一科为空，则返回“缺考”。

C2	A	B	C	D	E	F	G	H
	英语	语文	平均分					
1								
2	57	85	71					
3	72	85	78.5					
4		79	缺考					
5	77		缺考					
6	83	50	66.5					

我们根据题意，在 C2 单元格写出公式：=IF(OR(ISBLANK(A2), ISBLANK(B2)), “缺考”, AVERAGE(A2:B2))，并下拉复制完成。

## 名次排位函数——RANK 函数

rank 函数最常用的是求某一个数值在某一区域内的排名。

rank 函数语法形式：rank(number, ref, [order])

函数名后面的参数中 number 为需要求排名的那个数值或者单元格名称(单元格内必须为数字)，ref 为排名的参照数值区域，order 的为 0 和 1，默认不用输入，得到的就是从大到小的排名，若是想求倒数第几，order 的值请使用 1。

下面给出几个 rank 函数的范例：

示例 1：正排名

	B2			
			<b>=RANK(20,A1:A5)</b>	
	A	B	C	D
1	10			
2	15	1		
3	20			
4	14			
5	20			

此例中，我们在 B2 单元格求 20 这个数值在 A1:A5 区域内的排名情况，我们并没有输入 order 参数，不输入 order 参数的情况下，默认 order 值为 0，也就是从高到低排序。此例中 20 在 A1:A5 区域内的正排序是 1，所以显示的结果是 1。

示例 2：倒排名

	B2			
			<b>=RANK(20,A1:A5,1)</b>	
	A	B	C	D
1	10			
2	15	4		
3	20			
4	14			
5	20			

此例中，我们在上面示例的情况下，将 order 值输入为 1，发现结果大变，因为 order 值为 1，意思是求倒数的排名，20 在 A1:A5 区域内的倒数排名就是 4。

示例 3：求一列数的排名

	B1			
			<b>=RANK(A1,A\$1:A\$5)</b>	
	A	B	C	D
1	10	5		
2	15	3		
3	20	1		
4	14	4		
5	20	1		

在实际应用中，我们往往需要求某一列的数值的排名情况，例如，我们求 A1 到 A5 单元格内的数据的各自排名情况。我们可以使用单元格引用的方法来排名：

=rank(a1,a1:a5)，此公式就是求 a1 单元格在 a1:a5 单元格的排名情况，当我们使用自动填充工具拖拽数据时，发现结果是不对的，仔细研究一下，发现 a2 单元格的公式

居然变成了 =rank(a2, a2:a6) 这超出了我们的预期，我们比较的数据的区域是 a1:a5，不能变化，所以，我们需要使用 \$ 符号锁定公式中 a1:a2 这段公式，所以，a1 单元格的公式就变成了 =rank(a1, a\$1:a\$5)。

## 提取字符串中的某些字符函数——LEFT, RIGHT, MID,

LEFT, RIGHT, MID, 这三个函数都是文本的提取的。前两个的格式是一样的，只是提取的方向正好相反。LEFT 是从左向右取，RIGHT 是从右向左取。

LEFT (text, num\_chars)

RIGHT (text, num\_chars)

第一个参数 text，是文本，是你想要在那提取东西的文本，可以是一个字符串，或是一个单元格。第二个参数 chars 是想要提取的个数。

比如：=LEFT (A1, 2)，也是从 A1 单元格的文本里，从左边第一位开始，向右提取两位。比如 A1 的数据是：“找吃的蚂蚁”，得到的结果是：“找吃”。

那么：=RIGHT (A1, 2)，也就是从 A1 单元格的文本里，从右边第一位开，向左提取两位，这样就得到：“蚂蚁”。

当然你也可以这样写，如果你喜欢：=LEFT (“找吃的蚂蚁”， 2)。

MID 这个函数也是从左向右提取的，但不一定是从第一个起，可以从中间开始。先看它的格式吧：MID (text, start\_num, num\_chars)

第一个参数也是 text，它的属性与前面两个是一样的。第二个参数，star\_num，也就是要提取的开始字符，第三个参数 num\_chars 是要提取的个数。

用前面的例子得到“找吃”的结果的：=MID (A1, 1, 2)，而得到“蚂蚁”的公式是：=MID (A1, 4, 2)。

第一个公式的意思是：在 A1 单元格里提取文本，从文本的第 1 位开始，向右提取 2 个。


第二个公式的意思是：在 A1 单元格里提取文本，从文本的第 4 位开始，向右提取 2 个。

注意，MID 函数只能从左向右提取的。



下面以使用MID函数从身份证号码中提取出生年月为例，介绍MID函数的具体应用方法。

**01** 在工作表中输入如下图所示的示例数据。




The screenshot shows an Excel spreadsheet with columns A, B, and C. Column A is labeled '身份证号码' (ID Number) and contains three values: 511622198605127896, 510602186302061253, and 611402198306217893. Column B is labeled '出生日期' (Birth Date) and is currently empty. A yellow callout bubble with the text '输入' (Input) points to cell B2.

	A	B	C
1	身份证号码	出生日期	
2	511622198605127896		
3	510602186302061253		
4	611402198306217893		

**02** 选中B2单元格，在其中输入公式

"=MID(A2,7,4)&"年"&MID(A2,11,2)&"月""，然后按“Enter”键确认，即可从第一个身份证号码中提取出这个人的出生年月。



The screenshot shows the same Excel spreadsheet as before, but now cell B2 contains the result '1986年05月'. A yellow callout bubble with the text '1.选中' (1. Selected) points to cell B2, and another yellow callout bubble with the text '2.输入公式并确认' (2. Input formula and confirm) points to cell B2. The formula bar at the top shows the formula: =MID(A2,7,4)&"年"&MID(A2,11,2)&"月".

	A	B	C	D	E
1	身份证号码	出生日期			
2	511622198605127896	1986年05月			
3	510602186302061253				
4	611402198306217893				

## 插入当前时间函数——NOW 函数

TODAY 和 NOW 都是 Excel 中与日期时间相关的函数，二者的不同之处是：

TODAY 函数仅插入当前的日期，如：

=TODAY()

会在所输入的单元格中显示当前日期，如“2009/5/14”。

而 NOW 函数同时插入日期和时间，如：

=NOW()

单元格中显示为：“2009/5/14 16:06”

因此，当需要计算某日期距今天有多少整数天时，应使用 TODAY 函数。例如今天为 2009 年 5 月 14 日，计算上海世博会距今有多少天：

=DATE(2010, 5, 1)-TODAY()

公式返回 352 天。

TODAY函数用来取得当前日期的数值，其函数语法为TODAY()。

同NOW函数相比，使用这个函数只显示当前日期而不显示时间。TODAY的应用示例及说明如下图所示。

	A	B	C
1	公式	结果	说明
2	=TODAY()	2010/5/13	返回当前日期
3	=TODAY()+5	2010/5/18	返回当前日期加5天。本列中当前日期为2010/5/13，因此返回2010/5/18。
4	=DATEVALUE("2012/12/30")-TODAY()	962	返回当前日期和2012/12/30日之间的天数。请注意，A4单元格必须为“常规”或“数值”格式才能正确显示结果。
5	=DAY(TODAY())	13	返回一月中当前日期。
6	=MONTH(TODAY())	5	返回一年中的当前月分（1-12）。例如，本例中当前月份为5月，因此此公式返回“5”。

DATE函数用来将年、月和日组合成日期，其函数语法为DATE(year,month,day)。其中，year表示年份，month表示月份，day表示当前日期。

## 计算两个日期期间的天数——DATEVALUE 函数

DATEVALUE函数的主要功能是将以文本表示的日期转换成一个序列号。例如，公式=DATEVALUE("2008-1-1")返回日期2008-1-1的序列号39448。

**01** 将要输入日期的单元格格式设置为文本，然后输入如下图所示的示例数据。

	A	B	C	D
1	借书日期	还书日期	借书天数	
2	2010/4/25	2010/5/29		
3				
4				
5				
6				
7				
8				
9				
10				
11				
12				

**02** 选中C2单元格，在其中输入公式

"=DATEVALUE(B2)-DATEVALUE(A2)"，然后按下“Enter”键，即可计算出借书天数。

	A	B	C	D
1	借书日期	还书日期	借书天数	
2	2010/4/25	2010/5/29	34	
3				
4				
5				
6				
7				
8				
9				
10				
11				
12				
13				
14				
15				

## 分解时间函数——YEAR 函数

利用YEAR，MONTH和DAY函数可将人员的出生日期分解成年、月和日，其操作步骤如下。

**01** 在工作表中输入如下图所示的示例数据。

	A	B	C	D	E
1	出生日期	年	月	日	
2	1983年3月25日				
3	1988年5月25日				
4	1982年1月20日				
5	1979年9月9日				

输入

**02** 选中B2单元格，在其中输入公式“=YEAR(A2)”，然后按下“Enter”键，即可根据A2单元格中的出生日期计算出出生年份。

	A	B	C	D	E
1	出生日期	年	月	日	
2	1983年3月25日	1983			
3	1988年5月25日				
4	1982年1月20日				
5	1979年9月9日				

计算出生年份

**03** 选中C2单元格，在其中输入公式“=MONTH(A2)”，然后按下“Enter”键，即可根据A2单元格中的出生日期计

算出出生月份。

	A	B	C	D	E
1	出生日期	年	月	日	
2	1983年3月25日	1983	3		
3	1988年5月25日				
4	1982年1月20日				
5	1979年9月9日				

计算出生月份

**04** 选中D2单元格，在其中输入公式“=DAY(A2)”，然后按下“Enter”键，即可根据A2单元格中的出生日期计算出出生日。

	A	B	C	D	E
1	出生日期	年	月	日	
2	1983年3月25日	1983	3	25	
3	1988年5月25日				
4	1982年1月20日				
5	1979年9月9日				

计算出生日

**提示**

利用填充功能即可根据A列中的数据计算出其他人员的出生年份、月份和出生日。



## 计算员工工龄函数——YEAR 函数

**01** 在工作表中输入如下图所示的示例数据。

	A	B	C	D	E
1	姓名	入职日期	工龄		
2	陈国良	2008/5/20			
3	李思	2006/3/25			
4	涂方长	2005/2/24			
5					
6					
7					
8					
9					
10					
11					

**02** 选中C2单元格，在其中输入公式“=YEAR(TODAY())-YEAR(B2)”，然后按下“Enter”键，即可返回员工“陈国良”

的工龄。

	A	B	C	D	E
1	姓名	入职日期	工龄		
2	陈国良	2008/5/20	2		
3	李思	2006/3/25			
4	涂方长	2005/2/24			
5					
6					
7					
8					
9					
10					
11					
12					

**注意**

将C列单元格中的单元格格式设置为“常规”才能正确显示计算结果。

## 去尾取整函数——INT 函数

INT函数的应用示例如下图所示。

	A	B	C	D
1	数值	公式	结果	
2	5.5	=INT(A2)	5	
3	2.3	=INT(A3)	2	
4	-2.6	=INT(A4)	-3	
5	-3.1	=INT(A5)	-4	
6				
7				
8				
9				
10				
11				

注：不可调整小数位数。

## 向下舍数字函数——ROUNDDOWN 函数

(1)功能 按指定的位数对数值进行舍入。

(2)格式 ROUNDDOWN(数值或数值单元格, 指定的位数)

(3)示例

12.351					
325.525					
①保留2位小数——舍去千分位及以后的小数位保留到百分位。					
=ROUNDDOWN(A1, 2)	12.35				
=ROUNDDOWN(A2, 2)	325.52				
②舍去小数位保留整数——舍去十分位及以后的小数位保留整数					
=ROUNDDOWN(A1, 0)	12				
=ROUNDDOWN(A2, 0)	325				
③整数保留到十位——整数部分舍去个位上大于0的数字（用0					
=ROUNDDOWN(A1, -1)	10				
=ROUNDDOWN(A2, -1)	320				

说明：

函数 ROUNDDOWN 的第 1 个参数可以是具体的数值也可以是数值单元格引用。

函数 ROUNDDOWN 的第 2 个参数——指定保留的位数，保留小数位用正整数表示，即 1，2，3，4……（对应十分位、百分位、千分位、万分位……）；保留整数位用非正整数表示，即 0，-1，-2，-3，……（对应个位、十位、百位……）。

函数 ROUND 与函数 ROUNDDOWN 的对比：

ROUND 函数按指定位数把右侧数位上的数字进行四舍五入，

ROUNDDOWN 函数按指定位数把右侧数位上的数字舍弃为 0。

## 余数函数——MOD 函数

MOD（求余）函数用来返回两数相除的余数，结果的正负号与除数相同。例如，分子为10，分母为3，公式为10/3，用10除以3得到整除数3，余数1。因此，在求平均分配定量时的余数，或在预算范围内求购入商品后的余额时可使用MOD函数。

E2		=MOD(C2, B2)			
	A	B	C	D	E
1	商品名	单价	预算	预计购买数	余额
2	面包	1.8	1200	666	1.2
3	纯牛奶	2.5	2000	800	
4	豆奶	1.5	1000	666	
5					
6					
7					
8					

## 四舍五入函数——ROUND 函数

ROUND ( 四舍五入 ) 函数用来返回某个数字按指定位数取整后的数字，其函数语法为ROUND ( number,num\_digigs )。其中，函数参数number是要进行四舍五入的数字或表达式，num\_digits用来指定由第几位小数以下四舍五入。

	M16		f <sub>x</sub>		
	A	B	C	D	E
1	数值	公式	结果		
2	8.259	=ROUND(A2,0)	8		
3		=ROUND(A2,1)	8.3		
4		=ROUND(A2,-1)	10		
5		=ROUND(A2,2)	8.26		
6		=ROUND(A2,-2)	0		
7					
8					

## 查找函数

表 8-1 查找公式中使用的函数

函 数	说 明
CHOOSE	根据提供为参数的一个列表值(最多 29 个)返回指定的值
HLOOKUP	水平查找。在表格的顶行中查找一个值，并在该表格的指定行的相应列中返回一个值
INDEX	返回一个表格或范围中的一个值(或者对这个值的引用)
LOOKUP	从一个行范围或列范围中返回一个值。LOOKUP 函数的另一种形式类似于 VLOOKUP，但只能从范围的最后一列中返回一个值
MATCH	返回范围中与某个指定值匹配的项的相对位置
OFFSET	返回一个范围的引用，它是单元格或单元格范围中的指定行和列的编号
VLOOKUP	垂直查找。查找表格第一列中的某个值，并在该表格的指定列的相应行中返回一个值

## 寻找并参照函数——vlookup 函数（只查找第一次出现的）

首先我们介绍下使用的函数 vlookup 的几个参数，vlookup 是判断引用数据的函数，它总共有四个参数，依次是：

- 1、判断的条件
- 2、跟踪数据的区域
- 3、返回第几列的数据
- 4、是否精确匹配

## VLOOKUP 函数的使用方法（入门级）

VLOOKUP 是一个查找函数，给定一个查找的目标，它就能从指定的查找区域中查找返回想要查找到的值。它的基本语法为：

VLOOKUP (查找目标, 查找范围, 返回值的列数, 精确 OR 模糊查找)

下面以一个实例来介绍一下这四个参数的使用

例 1: 如下图所示, 要求根据表二中的姓名, 查找姓名所对应的年龄。

B13	=VLOOKUP(A13,\$B\$2:\$D\$8,3,0)			
A	B	C	D	E
1	表一:			
2	编号	姓名	性别	年龄
3	1	刘原	男	20
4	2	南星	男	16
5	3	吴路	女	30
6	4	张新远	男	20
7	5	李玉成	男	45
8	6	张欣	女	30
9				
10				
11	表二:			
12	姓名	年龄		
13	南星	16		
14	李玉成	45		
15				

公式: B13 =VLOOKUP(A13, \$B\$2:\$D\$8, 3, 0)

参数说明:

1 **查找目标**：就是你指定的查找的内容或单元格引用。本例中表二 A 列的姓名就是查找目标。我们要根据表二的“姓名”在表一中 A 列进行查找。

公式: B13 =VLOOKUP(A13, \$B\$2:\$D\$8, 3, 0)

2 查找范围 (VLOOKUP(A13, **\$B\$2:\$D\$8**, 3, 0) )：指定了查找目标，如果没有说从哪里查找，EXCEL 肯定会很为难。所以下一步我们就要指定从哪个范围中进行查找。VLOOKUP 的这第二个参数可以从一个单元格区域中查找，也可以从一个常量数组或内存数组中查找。本例中要从表一中进行查找，那么范围我们要怎么指定呢？这里也是极易出错的地方。大家一定要注意，给定的第二个参数查找范围要符合以下条件才不会出错：

A 查找目标一定要在该区域的第一列。本例中查找表二的姓名，那么姓名所对应的表一的姓名列，那么表一的姓名列（列）一定要是查找区域的第一列。象本例中，给定的区域要从第二列开始，即**\$B\$2:\$D\$8**，而不能是**\$A\$2:\$D\$8**。因为查找的“姓名”不在**\$A\$2:\$D\$8** 区域的第一列。

B 该区域中一定要包含要返回值所在的列，本例中要返回的值是年龄。年龄列（表一的 D 列）一定要包括在这个范围内，即：**\$B\$2:\$D\$8**，如果写成**\$B\$2:\$C\$8** 就是错的。

3 返回值的列数 (B13 =VLOOKUP(A13, **\$B\$2:\$D\$8**, **3**, 0) )。这是 VLOOKUP 第 3 个参数。它是一个整数值。它怎么得来的呢。它是“返回值”在第二个参数给定的区域中的列数。本例中我们要返回的是“年龄”，它是第二个参数**查找范围****\$B\$2:\$D\$8** 的第 3 列。这里一定要注意，列数不是在工作表中的列数（不是第 4 列），而是在查找范围区域的第几列。如果本例中要是查找姓名所对应的性别，第 3 个参数的值应该设置为多少呢。答案是 2。因为性别在**\$B\$2:\$D\$8** 的第 2 列中。

4 精确 OR 模糊查找 (VLOOKUP(A13, **\$B\$2:\$D\$8**, 3, **0**) )，最后一个参数是决定函数精确和模糊查找的关键。精确即完全一样，模糊即包含的意思。第 4 个参数如果指定值是 **0 或 FALSE** 就表示精确查找，而值为 **1 或 TRUE** 时则表示模糊。这里兰色提醒大家切记切记，在使用 VLOOKUP 时千万不要把这个参数给漏掉了，如果缺少这个参数默认为值为模糊查找，我们就无法精确查找到结果了。



好了，关于 VLOOKUP 函数的入门级应用就说到这里，VLOOKUP 函数可不只是这么简单的查找，我们讲的还只是 1/10 的用法。其他的没法在一篇文章中说明。敬请期待“VLOOKUP 的使用方法-进阶篇”吧。

## VLOOKUP 函数的使用方法（初级篇）

上一讲咱们学习了 VLOOKUP 的基本用法和示例，本讲将介绍 VLOOKUP 在使用中的一些小技巧。

### 一、VLOOKUP 多行查找时复制公式的问题

VLOOKUP 函数的第三个参数是查找返回值所在的列数，如果我们需要查找返回多列时，这个列数值需要一个个的更改，比如返回第 2 列的，参数设置为 2，如果需要返回第 3 列的，就需要把值改为 3。。。如果有十几列会很麻烦的。那么能不能让第 3 个参数自动变呢？向后复制时自动变为 2，3，4，5。。。。

在 EXCEL 中有一个函数 COLUMN，它可以返回指定单元格的列数，比如

=COLUMNS (A1) 返回值 1

=COLUMNS (B1) 返回值 2

而单元格引用复制时会自动发生变化，即 A1 随公式向右复制时会变成 B1, C1, D1。。。这样我们用 COLUMN 函数就可以转换成数字 1, 2, 3, 4。。。。

例：下例中需要同时查找性别，年龄，身高，体重。



B、查找的字符串或被查找的字符中含有空格或看不见的空字符，验证方法是用=号对比一下，如果结果是 FALSE，就表示两个单元格看上去相同，其实结果不同。

C、参数设置错误。VLOOKUP 的最后一个参数没有设置成 1 或者没有设置掉。第二个参数数据源区域，查找的值不是区域的第一列，或者需要返回的字段不在区域里，参数设置在入门讲里已注明，请参阅。

D、数值格式不同，如果查找值是文本，被查找的是数字类型，就会查找不到。解决方法是把查找的转换成文本或数值，转换方法如下：

文本转换成数值：\*1 或--或/1

数值转抱成文本：&"

VLOOKUP 函数的初级篇就说到这里了，咱们下一讲将介绍 VLOOKUP 的模糊查找有、反向查找等。敬请大家关注：VLOOKUP 函数的使用方法（进阶篇）。本系列为兰色幻想原创。转载请注明作者和转自“EXCEL 精英培训”

## **VLOOKUP 函数的使用方法（进阶篇）**

在学习了 VLOOKUP 的入门和初级篇后，本文将带将大家学习 VLOOKUP 的进阶篇：VLOOKUP 的模糊查找。

### **一、字符的模糊查找**

在 A 列我们知道如何查找型号为“AAA”的产品所对应的 B 列价格，即：  
=VLOOKUP (C1, A:B, 2, 0)

如果我们需要查找包含“AAA”的产品名称怎么表示呢？如下图表中所示。

	B10		=VLOOKUP("*"&A10&"*",A2:B6,2,0)			
	A	B	C	D	E	F
1	产品	价格				
2	AAB8-32	1				
3	BRE-533	3				
4	SFEW-324	5				
5	AAA-22	6				
6	WEWEE-22	1				
7						
8						
9	产品	价格				
10	AAA	6				
11						

公式=VLOOKUP("\*"&A10&"\*",A2:B6,2,0)

**公式说明：**VLOOKUP 的第一个参数允许使用通配符“\*”来表示包含的意思，把\*放在字符的两边，即“\*” & 字符 & “\*”。

## 二、数字的区间查找

数字的区间查找即给定多个区间，指定一个数就可以查找出它在哪个区间并返回这个区间所对应的值。

在 VLOOKUP 入门中我们提示 VLOOKUP 的第 4 个参数，如果为 0 或 FALSE 是精确查找，如果是 1 或 TRUE 或省略则为模糊查找，那么实现区间查找正是第 4 个参数的模糊查找应用。

首先我们需要了解一下 VLOOKUP 函数模糊查找的两个重要规则：

1、引用的数字区域一定要从小到大排序。杂乱的数字是无法准确查找到的。如下面 A 列符合模糊查找的前题，B 列则不符合。

	A	B	
1	2	34	
2	34	67	
3	67	2	
4	78	54	
5	99	89	
6	符合条件	不符合条件	
7			

2、模糊查找的原理是：给一定个数，它会找到和它最接近，但比它小的那个数。详见下图说明。

	A	B	C	D
1	2			
2	34			
3	67			
4	78			
5	99			
6				
7				
8				
9				

查找77时会找到67，因为和77最接近的有两个数，一个78，一个67，但67比77小，所以查找77会找到67而不是78

最后看一个实例：

例：如下图所示，要求根据上面的提成比率表，在提成表计算表中计算每个销售额的提成比率和提成额。

B11		=VLOOKUP(A11,\$A\$3:\$B\$7,2)		
	A	B	C	D
1	提成比率表			
2	销售额	比率	表示区间	
3	0	1%	大于等于0小于10000	
4	10000	2%	大于等于10000小于30000	
5	30000	3%	大于等于30000小于40000	
6	40000	4%	大于等于40000小于50000	
7	200000	5%	大于200000	
8				
9	提成计算表			
10	销售额	提成比率	提成	
11	5000	0.01	50	
12	56500	0.04	2260	
13	12900	0.02	258	
14	500000	0.05	25000	
15				

公式：=VLOOKUP(A11,\$A\$3:\$B\$7,2)

公式说明：

1、上述公式省略了 VLOOKUP 最后一个参数，相当于把第四个参数设置成 1 或 TRUE。  
这表示 VLOOKUP 要进行数字的区间查找。

2、图中公式中在查找 5000 时返回比率表 0 所对应的比率 1%，原因是 0 和 10000 与 5000 最接近，但 VLOOKUP 只选比查找值小的那一个，所以公式会返回 0 所对应的比率 1%。

## VLOOKUP 函数的使用方法(高级篇)

前言：前面我们分别学习了 VLOOKUP 函数的入门、初级和进阶篇。今天我们学习 VLOOKUP 函数的高级应用部分-VLOOKUP 函数的数组应用。

### 一、VLOOKUP 的反向查找。

一般情况下，VLOOKUP 函数只能从左向右查找。但如果需要从右向右查找，则需要把区域进行“乾坤大挪移”，把列的位置用数组互换一下。

例 1：要求在如下图所示表中的姓名反查工号。

	B9		=VLOOKUP(A9, IF({1,0}, B2:B5, A2:A5), 2, 0)				
	A	B	C	D	E	F	G
1	工号	姓名					
2	A001	张一					
3	A002	赵三					
4	A003	杨五					
5	A004	孙二					
6							
7							
8	姓名	工号					
9	赵三	A002					
10							
11							

公式：=VLOOKUP(A9, IF({1,0}, B2:B5, A2:A5), 2, 0)

公式剖析：

1、这里其实不是 VLOOKUP 可以实现从右至左的查找，而是利用 IF 函数的数组效应把两列换位重新组合后，再按正常的从左至右查找。

2、IF({1,0}, B2:B5, A2:A5)这是本公式中最重要的组成部分。在 EXCEL 函数中使用数组时（前提是该函数的参数支持数组），返回的结果也会是一个数组。这里 1

和 0 不是实际意义上的数字，而是 1 相关于 TRUE，0 相当于 FALSE，当为 1 时，它会返回 IF 的第二个参数（B 列），为 0 时返回第二个参数（A 列）。根据数组运算返回数组，所以使用 IF 后的结果返回一个数组（非单元格区域）：“张一”，“A001”；“赵三”，“A002”；“杨五”，“A003”；“孙二”，“A004”}

## 二、VLOOKUP 函数的多条件查找。

VLOOKUP 函数需要借用数组才能实现多条件查找。

例 2：要求根据部门和姓名查找 C 列的加班时间。

C9		{=VLOOKUP(A9&B9, IF({1,0}, A2:A5&B2:B5, C2:C5), 2, 0)}						
	A	B	C	D	E	F	G	H
1	部门	姓名	加班时间					
2	销售	张一	1					
3	销售	赵三	5					
4	人事	杨五	3					
5	销售	赵三	6					
6								
7								
8	工号	姓名	加班时间					
9	销售	赵三	5					
10								

分析：我们可以延用例 1 的思路，我们的努力方向不是让 VLOOKUP 本身实现多条件查找，而是想办法重构一个数组。多个条件我们可以用&连接在一起，同样两列我们也可以连接成一列数据，然后用 IF 函数进行组合。

公式：{=VLOOKUP(A9&B9, IF({1,0}, A2:A5&B2:B5, C2:C5), 2, 0)}

公式剖析：

1、A9&B9 把两个条件连接在一起。把他们做为一个整体进行查找。

2、A2:A5&B2:B5，和条件连接相对应，把部门和姓名列也连接在一起，作为一个待查找的整体。

3、IF({1,0}, A2:A5&B2:B5, C2:C5) 用 IF({1,0}) 把连接后的两列与 C 列数据合并成一个两列的内存数组。按 F9 后可以查看的结果为：

“销售张一”，1；“销售赵三”，5；“人事杨五”，3；“销售赵三”，6}



4、完成了数组的重构后，接下来就是 VLOOKUP 的基本查找功能了，另外公式中含有多个数据与多个数据运算(A2:A5&B2:B5)，，所以必须以数组形式输入，即按 ctrl+shift 后按 ENTER 结束输入。

### 三、VLOOKUP 函数的批量查找。

VLOOKUP 一般情况下只能查找一个，那么多项该怎么查找呢？

例 3 要求把如图表中所有张一的消费金额全列出来

	A	B	C
1	日期	姓名	消费额
2	2011-2-1	张一	100
3	2011-2-2	王二	2
4	2011-2-3	张一	200
5	2011-2-4	李三	44
6	2011-2-5	张一	1000
7			
8			
9		张一	100
10			200
11			1000
12			

分析：经过前面的学习，我们也有这样一个思路，我们在实现复杂的查找时，努力的方向是怎么重构一个查找内容和查找的区域。要想实现多项查找，我们可以对查找的内容进行编号，第一个出现的是后面连接 1，第二个出现的连接 2。。。

公式：

```
=VLOOKUP(B$9&ROW(A1), IF({1,0}, $B$2:$B$6&COUNTIF(INDIRECT("b2:b"&ROW($2:$6)), B$9), $C$2:$C$6), 2, )}
```

公式剖析：

- 1、B\$9&ROW(A1) 连接序号，公式向下复制时会变成 B\$9 连接 1，2，3
- 2、给所有的张一进行编号。要想生成编号，就需要生成一个不断扩充的区域 (INDIRECT("b2:b"&ROW(\$2:\$6))), 然后在这个逐行扩充的区域内统计“张一”的个数，在连接上 \$B\$2:\$B\$6 后就可以对所有的张一进行编号了。

- 3、IF ({1, 0}) 把编号后的 B 列和 C 组重构成一个两列数组

通过以上的讲解，我们需要知道，VLOOKUP 函数的基本用法是固定的，要实现高级查找，就需要借助其他函数来重构查找内容和查找数组。



## 返回指定行、列或单元格的值——INDEX 函数

INDEX(数据区域(可以有多个数据区域), 返回第几行数据, 返回第几列数据, 查找第几个区域的数据)

INDEX 函数的功能就是返回指定单元格区域或数组常量。如果同时使用参数行号和列号, 函数 INDEX 返回 行号和列号交叉处的单元格中的值。

**INDEX 函数语法:** INDEX(单元格区域, 行号, 列号)

可以通过下图来认识 INDEX 函数的用法:

	A	B	C	D
1	78	87	76	
2	80	97	84	
3	91	75	64	
4	88	86	68	
5	93	99	83	
6	89	69	79	
7				
8	结果		公式	
9	84		=INDEX(A1:C6, 2, 3)	

=INDEX(A1:C6, 2, 3), 意思就是返回 A1:C6 中行号是 2 列号是 3 , 即第二行与第三列的交叉处, 也就是 C2 单元格的值, 为 84。

### 二、实例介绍

看下表; 下表中, 有两个数据区域, 第一个区域为 A2: C6; 第二个区域为 A8: C10;

	A	B	C
1	姓名	部门	年龄
2	Mary	外贸组	23
3	lucy	管理部门	24
4	LiLi	公关	25
5	陆青山	总裁	29
6	哈利	技术总监	22
7			
8	Aini	秘书	25
9	小昆虫	副总裁	29
10	Bier	投资方	56

现在, 我们将要使用 INDEX 函数, 求取这两个数据区域中的某单元格的数据。

下面, 我们要获取 C9 单元格的值, 即 29, 如何通过这个函数来获取数据呢?

还是先从函数语法来分析:

INDEX(数据区域, 返回第几行数据, 返回第几列数据, 查找第几个区域的数据)

通过套用以上的函数语法, 我们应该使用如下的函数公式:

=INDEX((A2:C6, A8:C10, A12:C13), 2, 3, 2)

E5	=INDEX((A2:C6, A8:C10), 2, 3, 2)				
	A	B	C	D	E
1	姓名	部门	年龄		
2	Mary	外贸组	23		
3	lucy	管理部门	24		
4	LiLi	公关	25		
5	陆青山	总裁	29		返回值 29
6	哈利	技术总监	22		
7					
8	Aini	秘书	25		
9	小昆虫	副总裁	29		
10	Bier	投资方	56		

如上图吧, 输入公式之后, 按下回车键即可得结果。

### 三、函数公式分析

=INDEX((A2:C6, A8:C10), 2, 3, 2), 其中:

(A2:C6, A8:C10)代表函数要查找的数据范围为两个区域, 中间使用, 隔开, 第一个区域为 A2:C6; 第二个区域为 A8:C10; (如果有第三个、第四个区域, 同样可以填入函数中)。

=INDEX((A2:C6, A8:C10), 2, 3, 2), 其中的 2 代表要查找的数据范围为第二个数据区域, 即 A8:C10; 而不是第一个区域 A2:C6;

=INDEX((A2:C6, A8:C10), 2, 3, 2), 其中的 3 代表返回已选择的数据区域的第三列数据;

=INDEX((A2:C6, A8:C10), 2, 3, 2), 其中的 2 代表返回已选择的数据区域的第二行数据;

当然, 该函数也可以只使用一个数据区域。比如, 同样, 要返回 C9 单元格的值, 我们可以使用如此公式代码:

=INDEX((A8:C10), 2, 3)

此时, 由于只有一个数据区域, 因此, 数据区域的参数可省略不填写。

### 四: 知识扩展

该函数的使用极其灵活, 因为任何一个参数都可变化。

数据范围: 可以只有一个数据范围, 也可以有多个数据范围(多个使用逗号隔开); 要返回的数据的行和列号, 自行设定。

要返回哪个数据范围的数据，就填写序号，比如返回第一个数据范围的数据，就填写 1，其它依此类推。

当然了，如果您使用该函数发生错误时，只能说明您指定的参数存在错误。

## 返回匹配值的位置函数——MATCH 函数(只查找第一次出现的)

MATCH 函数也是一个查找函数。MATCH 函数会返回匹配值的位置而不是匹配值本身。在使用时，MATCH 函数在众多的数字中只查找第一次出现的，后来出现的它返回的也是第一次出现的位置。

**MATCH 函数语法：**MATCH(查找值，查找区域，查找模式)

可以通过下图来认识 MATCH 函数的用法：

	A	B
1	<b>Product</b>	<b>Count</b>
2	Bananas	25
3	Oranges	38
4	Apples	40
5	Pears	41

=MATCH(41, B2:B5, 0)，得到结果为 4，返回数据区域 B2:B5 中 41 的位置。

=MATCH(39, B2:B5, 1)，得到结果为 2，由于此处无正确匹配，所以返回数据区域 B2:B5 中 (38) 的位置。注：匹配的查找值，MATCH 函数会查找小于或等于 (39) 的最大值。

=MATCH(40, B2:B5, -1)，得到结果为 #N/A，由于数据区域 B2:B5 不是按降序排列，所以返回错误值。

如果 match\_type 为 1，函数 MATCH 查找小于或等于 lookup\_value 的最大数值。

Lookup\_array 必须按升序排列：...、-2、-1、0、1、2、...、A-Z、FALSE、TRUE。

如果 match\_type 为 0，函数 MATCH 查找等于 lookup\_value 的第一个数值。

Lookup\_array 可以按任何顺序排列。

如果 match\_type 为 -1，函数 MATCH 查找大于或等于 lookup\_value 的最小数值。  
Lookup\_array 必须按降序排列：TRUE、FALSE、Z-A、...、2、1、0、-1、-2、...，等等。

## INDEX 和 MATCH 函数组合应用（相当于 vlookup 函数）

下图工作表所示的是一个产品的型号和规格的价格明细表。通过这个表的数据，进行一些对应的查询操作。

	D	E	F	G
3	规格 型号	101	201	301
4	A0110	78	87	76
5	A0111	80	97	84
6	A0112	91	75	64
7	A0113	88	86	68
8	A0114	93	99	83
9	B1120	89	69	79
10	B1121	91	70	69
11	B1122	77	91	81
12	B1123	98	75	74

第一，单击 B5 单元格下拉按钮，选择型号，然后在 B6 单元格完成型号所在行号的查询。如下图所示：

	A	B	C
5	查询型号		▼
6	型号所在行号		

随意选择一个型号，比如 A0110，然后在 B6 单元格输入公式：  
=MATCH(\$B\$5,\$D\$4:\$D\$12,0)，得到结果 1。

公式解释：用 MATCH 函数查找 B5 单元格这个型号在 D4:D12 区域中对应的位置。其中的 0 参数可以省略不写。MATCH 函数中 0 代表精确查找，1 是模糊查找。

第二，单击 B9 单元格下拉按钮，选择规格，然后在 B10 单元格完成规格所在列号的查询。如下图所示：

	A	B	C
9	查询规格		▼
10	规格所在列号		

随意选择一个规格，比如 101，然后在 B10 单元格输入公式：=MATCH(B9, E3:G3, 0)，得到结果 1。

第三，查询 B6 和 B10 单元格所对应的价格。

价格的查询，可以使用 index 函数完成，输入公式：=INDEX(E4:G12, B6, B10) 可以得到结果为 78。嵌套上面的 match 函数，可以将公式改为：  
=INDEX(E4:G12, MATCH(B5, D4:D12, 0), MATCH(B9, E3:G3, 0))。大家可以变化 C3 中的型号来看看结果是否正确。

通过下面工作表的源数据，利用 index 函数实现行列汇总查询。

	B	C	D	E	F	G	H
1	行列汇总：			规格	101	201	301
2				A0110	78	87	76
3	查找值：	<input type="text"/>		A0111	80	97	84
4	查找行号：			A0112	91	75	64
5	行汇总：			A0113	88	86	68
6				A0114	93	99	83
7	查找值：			B1120	89	69	79
8	查找列号：			B1121	91	70	69
9	列汇总：			B1122	77	91	81
10				B1123	98	75	74

INDEX 函数的帮助文件知道，如果将 row\_num 或 column\_num 设置为 0，函数 INDEX 则分别返回整个列或行的数组数值。通过用法说明，来实现上图的行列汇总。

首先，单击 C3 单元格下拉按钮，选择数据，比如选择 A0111，然后在 C4 单元格进行 C3 单元格对应的行号查找，公式为：=MATCH(C3, E2:E10,)，得到结果为 2，说明 A0111 在 E2:E10 区域的第二行。

然后对行号所对应的价格进行汇总求和。在 C5 单元格输入公式：  
=SUM(INDEX(F2:H10, MATCH(C3, E2:E10, ), ))

如果选中公式中的 INDEX(F2:H10, MATCH(C3, E2:E10, ), ) 部分，按下 F9 键，执行结果是 {80, 97, 84}，可以看到就是对第二行的数据进行求和。

同样的方法，可以进行列汇总，在 C9 单元格输入公式：  
=SUM(INDEX(F2:H10, , MATCH(C7, F1:H1, )))。

说明：查找行号和查找列号，只是过渡一下，帮助新手朋友加深对 match 函数的理解和使用，对熟悉的朋友，可以直接在行汇总一步完成公式的输入。

根据下面的工作表，进行区域汇总求和。

	B	C	D	E	F	G	H
15	区域汇总:			规格	101	201	301
16	开始行号:	<input type="text"/>		型号			
17	结束行号:			A0110	88	86	68
18	开始列号:			A0111	80	97	84
19	结束列号:			A0112	78	87	76
20	区域汇总:			A0113	93	99	83
21				A0114	91	75	64
22				B1120	89	69	79
23				B1121	91	70	69
24				B1122	77	91	81
				B1123	98	75	74

首先，分别在开始行号、结束行号、结束列号、结束列号选定需要求和的区域，比如 A0110、A0111、201、301。此题可以套用下图的格式进行求和。

**SUM (INDEX (区域, 开始行号, 开始列号) : (INDEX (区域, 结束行号, 结束列号))**

**G16**

**H17**

因此，在 C20 单元格输入求和公式：

=SUM (INDEX (F16:H24, MATCH (C16, E16:E24, ), MATCH (C17, E16:E24, )) : INDEX (F16:H24, MATCH (C18, F15:H15, ), MATCH (C19, F15:H15, )))



## 引用函数——OFFSET



	B	C	D	E	F	G	H
1	<b>函数解析：</b>						
2	不少朋友看到Offset有5个参数，第一反应就是太过复杂，其实这5个参数我们可以分成3组来看待。						
3	<b>=OFFSET(C13, 1, 2, 1, 1)</b>						
4	首先，Offset是个用来选择某一单元格区域的函数。在上面的公式里，我们用红、蓝、黑三种颜色把offset的五个参数分成了3组。						
5	其中红色参数（也就是第一个参数）代表执行这个函数的时候光标的起点，也就是C13单元格。						
6	蓝色的一组参数表示把起始单元格下移N行（参数2）和右移N列（参数3），本案例中就是从C13单元格开始，先下移1行，再右移2列，到达了E14单元格。						
7	最后一组的两个黑色参数，代表一个N行M列（参数4是N，参数5是M）的数据区域。这个区域是以前3个参数决定的光标的新位置（E14）为起点的，从这个单元格（E14）开始计算的N行M列（本案例参数4参数5都是1，所以是以E14为左上角的一行一列，也就是E14本身）。						
8	综合5个参数的效果，上面的函数的返回结果应该是9（E14单元格的值）。						
9		1	2	3	4	5	6
10		7	8	9	10	11	12
11		13	14	15	16	17	18
12		19	20	21	22	23	24
13							
14	<b>具体举例：（红色区域为函数返回结果）</b>						
15	<b>=OFFSET(C20, 1, 3, 1, 1)</b>						
16		1	2	3	4	5	6
17		7	8	9	10	11	12
18		13	14	15	16	17	18
19		19	20	21	22	23	24
20							
21	<b>=OFFSET(C26, 2, 3, 2, 1)</b>						
22		1	2	3	4	5	6
23		7	8	9	10	11	12
24		13	14	15	16	17	18
25		19	20	21	22	23	24
26							
27	<b>=OFFSET(C32, 0, 2, 3, 4)</b>						
28		1	2	3	4	5	6
29		7	8	9	10	11	12
30		13	14	15	16	17	18
31		19	20	21	22	23	24
32							

33	<b>变动方法:</b>					
34	通过上面三个案例，我们发现Offset的返回结果可以是一个数据区域。正常情况下，我们是可以用鼠标选中一个连续区域去绘制图表的，这也就使Offset的返回结果可以成为图表的数据源。					
35	接下来，我们只要通过调整Offset参数的数值，就可以随时改变该参数引用的区域了。					
36	比如下面的例子，我们将Offset的第2、3个参数制定在D44和F44单元格上，那么我们只要随时改变单元格的数值，就可以改变选中区域（即红色区域）了。					
37						
38						
39						
40	参数2:	1	参数3:	2		
41	公式	<b>=OFFSET(C46,D44,F44,1,1)</b>			实际结果:	9
42	1	2	3	4	5	6
43	7	8	9	10	11	12
44	13	14	15	16	17	18
45	19	20	21	22	23	24
46						
47	当如上面案例所示时，参数2等于1，参数3等于2，那么实际结果为9，即从C46开始下移一行右移两行以后的1行1列的区域。如果我们将D44和F44分别改称3和1，发现实际结果会改为20。这时候，通过更改函数参数来更改选择区域的效果就实现了。					
48						
49	如果上面的变化原理你理解了，那么恭喜你，最基本的变动图表原理你已经掌握了。下一步，我们就要学习如何把可变动的Offset函数返回值变成图表的过程了。					
50						
51						

OFFSET(基点单元格，移动的行数，移动的列数，所要引用的高度，所要引用的宽度)  
例

=OFFSET(A3,2,3,1,1)

A3 是基点单元格，

2 是正数，为向下移动 2 行，负为向上移动 2 行

3 是正数，为向右移动 3 列，负为向左移动 3 列

1 是引用 1 个单元格的高度

1 是引用 1 个单元格的宽度

它的结果是引用了 D5 中数值。

作用：以指定的引用为参照系，通过给定偏移量得到新的引用。返回的引用可以为一个单元格或单元格区域。并可以指定返回的行数或列数。

公式 说明（结果）

=OFFSET(C3, 2, 3, 1, 1) 显示单元格 F5 中的值 (0)

=SUM(OFFSET(C3:E5, -1, 0, 3, 3)) 对数据区域 C2:E4 求和 (0)

=OFFSET(C3:E5, 0, -3, 3, 3) 返回错误值 #REF!, 因为引用区域不在工作表中  
也可以运用 counta 函数看看有多少非空单元格=counta(OFFSET(C3:E5, -1, 0, 3, 3))

## OFFSET 函数的用法及应用实战讲解（上）

首先，认识一下 OFFSET 函数。

从下图说明来认识一下 excel 中 OFFSET 函数的用法。

	C7		=SUM(OFFSET(C2,1,2,3,1))						
	A	B	C	D	E	F	G	H	
1	1	2	3	4	5	6	7	8	
2	4	5	3	4	2	4	3	2	
3	5	3	5	6	3	5	4	3	
4	5	6	5	6	8	9	3	5	
5	6	9	5	4	7	9	6	3	
6	5	7	6	3	6	5	4	2	
7			18						

在 C7 单元格，输入公式：=SUM(OFFSET(C2,1,2,3,1))，得到结果为 18。这个公式是什么意思呢？就是计算 C2 单元格靠下 1 行并靠右 2 列的 3 行 1 列的区域的和。

可以在公式编辑栏，选中 OFFSET(C2,1,2,3,1) 部分，按 F9 键抹黑，得到运算结果为：{3;8;7}，此时公式变为：=SUM({3;8;7})。从上图可以得知，就是利用 OFFSET 函数来得到一个新的区域，然后使用 SUM 函数求出这个新区域的和。

下面，介绍 OFFSET 函数的用法。

Offset 函数主要应用在单元格区域的定位和统计方面，一般做数据透视表定义名称都需要用到 Offset 函数。Offset 函数属于查找与引用类的函数。

OFFSET 函数以指定的引用为参照系，通过给定偏移量得到新的引用。返回的引用可以为一个单元格或单元格区域，并可以指定返回的行数或列数。

OFFSET 函数的语法是：OFFSET(reference, rows, cols, height, width)，按照中文的说法即是：OFFSET(引用区域, 行数, 列数, [高度], [宽度])

其中的参数意义如下：

Reference：作为偏移量参照系的引用区域。Reference 必须为对单元格或相连单元格区域的引用；否则，函数 OFFSET 返回错误值 #VALUE!。

Rows：相对于偏移量参照系的左上角单元格，上(下)偏移的行数。如果使用 5 作为参数 Rows，则说明目标引用区域的左上角单元格比 reference 低 5 行。行数可为



正数(代表在起始引用的下方)或负数(代表在起始引用的上方)。

**Cols:** 相对于偏移量参照系的左上角单元格, 左(右)偏移的列数。如果使用 5 作为参数 Cols, 则说明目标引用区域的左上角的单元格比 reference 靠右 5 列。列数可为正数(代表在起始引用的右边)或负数(代表在起始引用的左边)。

**Height:** 高度, 即所要返回的引用区域的行数。Height 必须为正数。

**Width:** 宽度, 即所要返回的引用区域的列数。Width 必须为正数。

#### 学习使用 OFFSET 函数需要注意以下几点:

第一, 如果行数和列数偏移量超出工作表边缘, 函数 OFFSET 返回错误值 #REF!。

第二, 如果省略 height 或 width, 则假设其高度或宽度与 reference 相同。

第三, 函数 OFFSET 实际上并不移动任何单元格或更改选定区域, 它只是返回一个引用。

第四, 函数 OFFSET 可用于任何需要将引用作为参数的函数。

下面, 结合几个实例学习 OFFSET 函数的应用。

OFFSET 函数通常与其它函数来嵌套使用。下图所示的区域为实例的源数据区域。

	A	B	C
1	班级	姓名	成绩
2	1	张三	84
3	1	王五	75
4	1	李四	99
5	1	闻一	59
6	2	张二	65
7	2	江八	86
8	2	张六	93
9	3	赵六	82
10	3	钱子	75
11	3	赵一	66
12	3	孙六	52
13	3	周末	100

**OFFSET 函数应用第一题:** 如下图所示, 计算上图 B 列姓名下面所有单元格的数量。

F5			=COUNTA(OFFSET(B2,0,0,COUNTA(B:B)-1))			
	D	E	F	G	J	K
2			计算B列姓名下面的单元格数量:			
3			函数公式:			
4						
5			验证: 12			

在 F3 单元格输入函数: =OFFSET(B2,0,0,COUNTA(B:B)-1), 全选公式, 按 F9 键, 得到:={"张三";"王五";"李四";"闻一";"张二";"江八";"张六";"赵六";"钱子";"赵一";"孙六";"周末"}, 我们可以数一下个数, 得到 12 个。

提示: COUNTA 函数就是返回参数列表中非空值的单元格个数。

接下来, 在 F5 单元格输入公式: =COUNTA(OFFSET(B2,0,0,COUNTA(B:B)-1)), 得多结果为 12。

**OFFSET 函数应用第二题:** 如下图所示, 计算最后的平均成绩。分三种情况: 第一, 求最后一位、最后三位、最后五位的成绩。

F9			=AVERAGE(OFFSET(C1,COUNTA(C:C)-1,, -F8))			
	D	E	F	G	J	K
7			计算最后的平均成绩:			
8			最后个数: 5			
9			函数公式: 75.00			

单击 F8 单元格, 可以看到有一个向下的箭头, 单击箭头可以看到有 1、3、5 三个选项。

在 F9 单元格输入公式: =AVERAGE(OFFSET(C1,COUNTA(C:C)-1,, -F8)), 就可以得到 F8 单元格所显示的平均成绩。

公式分析: 首先使用 OFFSET 函数来计算出行数的值, 然后使用 AVERAGE 来除以 OFFSET 部分得到的值, 就得到了最后几个的平均成绩。

公式中, 其中 COUNTA(C:C) 部分是统计 C 列的非空单元格个数为 13, 从而可以得到 COUNTA(C:C)-1 的值为 12, 即从 C1 单元格向下偏移 12 行。

另外 -F8 的意思是: 减去 F8 单元格中的值。

当改变 F8 单元格的值, 就可以得到其余行数的平均成绩。

**OFFSET 函数应用第三题:** 如下图所示, 要求统计出班级=H3, 成绩>=I3 的人数。

I5										
	G	H	I	J	K	L	M	N	O	P
2		班级	成绩							
3		3	70							
4		统计:班级=H3,成绩>=I3的人数								
5		统计结果	3							

H3 和 I3 单元格的值是条件。单击这两个单元格，右下角有一个向下的三角形，单击此按钮，可以显示选择其余的条件值。比如上图所示，就是统计出班级为 3 版，成绩在 >=70 分的人数有几人。

本题的统计结果，可以在 I5 单元格输入公式：  
=COUNTIF(OFFSET(C1, MATCH(H3, A2:A13, 0),, COUNTIF(A2:A13, H3)), ">="&I3)，得到结果。

公式分析：OFFSET 的第二参数使用 MATCH 来精确查找位置。偏移的行数就由 MATCH(H3, A2:A13, 0) 来控制，高度由 COUNTIF(A2:A13, H3) 来控制。然后把 OFFSET 函数的值用 COUNTIF 来计算符合条件的有多少个。

## OFFSET 函数的经典应用（下）

在第 4 期讲座中，简单老师从零讲解了 OFFSET 函数的使用，本文最主要通过一个实例来讲解 OFFSET 函数在定义名称方面的运用。

下图所示的工作表命名为：名称实例，其中 A1: C13 区域为源数据，本讲座要完成的作业是通过源数据，使用 OFFSET 函数定义一个叫“AA”的名称，然后完成下图 1、2、3 个班级的实考人数、总分、最高分、优秀数、平均分的统计。

本题最核心的一个问题就是定义 AA 名称。

单击菜单“插入”——“名称”——“定义”，在“名称”框中输入名称“AA”，然后在下面的引用位置输入：=OFFSET(名称实例!\$A\$1, MATCH(名称实例!\$D17, 名称实例!\$A\$2:\$A\$13, ), 2, COUNTIF(名称实例!\$A\$2:\$A\$13, 名称实例!\$D17))

引用位置输入公式的含义，解释如下。把上面这个公式分成两部分来查看，就很容易了。

E17	=COUNTA(AA)								
	A	B	C	D	E	F	G	J	K
1	班级	姓名	成绩						
2	1	张三	84						
3	1	王五	75						
4	1	李四	99						
5	1	周六一	59						
6	2	张二	65						
7	2	江八	86						
8	2	张六	93						
9	3	赵六	82						
10	3	钱子	75						
11	3	赵一	69						
12	3	孙六	23						
13	3	周末	100						
15									
16				班级	实考人数	总分	最高分	优秀数	平均分
17				1	4				
18				2					
19				3					

第一部分：MATCH(名称实例!\$D17, 名称实例!\$A\$2:\$A\$13,)，这个值是 OFFSET 的第二个参数，使用 MATCH 来精确查找位置。代表偏移的行数就由 MATCH 部分的值来控制。这个公式的意思就是查找 D17 单元格的值在 A2: A13 区域中的位置。

MATCH 函数也是一个查找函数。MATCH 函数会返回中匹配值的位置而不是匹配值本身。在使用时，输入单值它就返回单值，输入多值就返回多值。MATCH 函数在众多的数字中只查找第一次出现的，后来出现的它返回的也是第一次出现的位置。MATCH 函数一般都和别的函数嵌套使用。

第二部分：COUNTIF(名称实例!\$A\$2:\$A\$13, 名称实例!\$D17)，这个值是 OFFSET 的第四个参数，高度就由 COUNTIF 的值来控制。

定义完名称之后，在 E17 单元格输入公式：=COUNTA(AA)，然后下拉，就可以求出各个班级的实考人数。

在 F17 单元格输入公式：=SUM(AA)，然后下拉，就可以求出各个班级的总分。

在 G17 单元格输入公式：=MAX(AA)，然后下拉，就可以求出各个班级的最分。

在 J17 单元格套用 COUNTIF 函数的公式：countif(区域，条件)，输入公式：=COUNTIF(AA,">=85")，然后下拉，就可以求出各个班级的优秀人数。在这个公式中，我们假定优秀人数的分数为大于等于 85 分。

在 K17 单元格输入公式：=AVERAGE(AA)，然后下拉，就可以求出各个班级的平均分。